In [1]:

```
# Define A
A=matrix(QQ,[[1, 4, 5, 0, 9],[3, -2, 1, 0, -1],[-1, 0, -1, 0, -1],[2, 3, 5, 1, 8]])
show(A)
```

Out[1]:

$$\begin{pmatrix} 1 & 4 & 5 & 0 & 9 \\ 3 & -2 & 1 & 0 & -1 \\ -1 & 0 & -1 & 0 & -1 \\ 2 & 3 & 5 & 1 & 8 \end{pmatrix}$$

In [2]:

```
# Make use of leading 1 in first row first column to reduce
# entries below to be zeros
# we take away three times of R1 from R2
# R2:=R2-3*R_1
A.add_multiple_of_row(1,0,-3)
show(A)
```

Out[2]:

$$\begin{pmatrix} 1 & 4 & 5 & 0 & 9 \\ 0 & -14 & -14 & 0 & -28 \\ -1 & 0 & -1 & 0 & -1 \\ 2 & 3 & 5 & 1 & 8 \end{pmatrix}$$

In [3]:

```
# we add R1 to R3
# R3:=R3+R_1
A.add_multiple_of_row(2,0,1)
show(A)
```

Out[3]:

$$\begin{pmatrix} 1 & 4 & 5 & 0 & 9 \\ 0 & -14 & -14 & 0 & -28 \\ 0 & 4 & 4 & 0 & 8 \\ 2 & 3 & 5 & 1 & 8 \end{pmatrix}$$

In [4]:

```
# we take away twice of R1 from R4
# R4:=R4-2*R_1
A.add_multiple_of_row(3,0,-2)
show(A)
```

Out[4]:

$$\begin{pmatrix} 1 & 4 & 5 & 0 & 9 \\ 0 & -14 & -14 & 0 & -28 \\ 0 & 4 & 4 & 0 & 8 \\ 0 & -5 & -5 & 1 & -10 \end{pmatrix}$$

In [5]:

```
# in order to make entry "-14" in second row second column
# a leading 1, we rescale
# we divide R2 by -14
# R2:=(-1/14)*R2
A.rescale_row(1,-1/14)
show(A)
```

Out[5]:

$$\begin{pmatrix} 1 & 4 & 5 & 0 & 9 \\ 0 & 1 & 1 & 0 & 2 \\ 0 & 4 & 4 & 0 & 8 \\ 0 & -5 & -5 & 1 & -10 \end{pmatrix}$$

In [6]:

```
# we make use of leading 1 in second row second column to reduce
# entries below to be zeros
# we take away four times of R2 from R3
# R3:=R3-4*R2
A.add_multiple_of_row(2,1,-4)
show(A)
```

Out[6]:

$$\begin{pmatrix} 1 & 4 & 5 & 0 & 9 \\ 0 & 1 & 1 & 0 & 2 \\ 0 & 0 & 0 & 0 & 0 \\ 0 & -5 & -5 & 1 & -10 \end{pmatrix}$$

In [7]:

```
# since one whole row of entirely zeros should be at the bottom,
# we interchange (swap) R3 and R4
# R3:=R4 and R4:=R3
A.swap_rows(2,3)
show(A)
```

Out[7]:

$$\begin{pmatrix} 1 & 4 & 5 & 0 & 9 \\ 0 & 1 & 1 & 0 & 2 \\ 0 & -5 & -5 & 1 & -10 \\ 0 & 0 & 0 & 0 & 0 \end{pmatrix}$$

In [8]:

```
# we make use of leading one in second row second cloumn to reduce
# entry below to be zero
# we add five times of R2 to R3
# R3:=R3+5*R2
A.add_multiple_of_row(2,1,5)
show(A)
```

Out[8]:

$$\begin{pmatrix} 1 & 4 & 5 & 0 & 9 \\ 0 & 1 & 1 & 0 & 2 \\ 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 0 & 0 \end{pmatrix}$$

In [9]:

```
# it is already in row-echelon form
# to make it to reduced echelon form, we need to perform Jordan
# we make use of leading 1 in second row second column to reduce
# entry above to be zero
# R1:=R1-4*R2
A.add_multiple_of_row(0,1,-4)
show(A)
```

Out[9]:

$$\begin{pmatrix} 1 & 0 & 1 & 0 & 1 \\ 0 & 1 & 1 & 0 & 2 \\ 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 0 & 0 \end{pmatrix}$$

In [10]:

```
# Now it is in reduced row-echelon form
# instead of doing it step by step, we can also do it directly
# in CoCalc to obtain reduced row-echelon form (rref)
A=matrix(QQ,[[1, 4, 5, 0, 9],[3, -2, 1, 0, -1],[-1, 0, -1, 0, -1],[2, 3, 5, 1, 8]])
show(A.rref())
```

Out[10]:

$$\begin{pmatrix} 1 & 0 & 1 & 0 & 1 \\ 0 & 1 & 1 & 0 & 2 \\ 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 0 & 0 \end{pmatrix}$$

In [0]:

```
```